# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Charles Mauch <cmauch@gmail.com>

pkgname=perl-net-ipv4addr
pkgver=0.10
pkgrel=14
pkgdesc="Perl/CPAN Module Net::IPv4Addr"
arch=("any")
depends=('perl')
url="https://search.cpan.org/dist/Net-IPv4Addr"
license=("GPL" "PerlArtistic")
source=("https://www.cpan.org/authors/id/F/FR/FRAJULAC/Net-IPv4Addr-$pkgver.tar.gz")
sha256sums=('3845de4f30b17c842b192cace8879dbb6214de9692cfa70f0aaf0981422a63fe')

build() {
  cd "$srcdir"/Net-IPv4Addr-$pkgver
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd "$srcdir"/Net-IPv4Addr-$pkgver
  make install DESTDIR="$pkgdir"
  find "$pkgdir" -name '.packlist' -delete
  find "$pkgdir" -name '*.pod' -delete
}
